

SQL Functions can be very useful for solving complex problems - with one caveat, we must always consider the cost to SQL Server's performance. That being said the following is a SQL function that will take a date and work out the Financial Year, Quarter, and Month, it will also return a 'normalized' Financial date although this is not necessarily needed as we could always create this outside the function.

```

create function [dbo].[FinancialYQM] (
    @date datetime
, @offset int
, @monthEnd int
)
returns table
as
return
select
    Fin_Year
, Fin_Quarter
, Fin_Month
, convert(datetime, rtrim(t.Fin_Year) + '-' + rtrim(t.Fin_Month) + '-1') as
Fin_Date
from (
select
    datepart(year, dateadd(month, @offset, @date)) +
    case
        when datepart(month, dateadd(month, @offset, @date)) = 12 and datepart(day,
dateadd(month, @offset, @date)) > @monthEnd then 1
        else 0
    end as Fin_Year
, ceiling(convert(decimal,
    (datepart(month, dateadd(month, @offset, @date)) +
    case
        when datepart(month, dateadd(month, @offset, @date)) = 12 and datepart(day,
dateadd(month, @offset, @date)) > @monthEnd then -11
        when datepart(day, dateadd(month, @offset, @date)) > @monthEnd then 1
        else 0
    end)
) / 3) as Fin_Quarter
, datepart(month, dateadd(month, @offset, @date)) +
    case
        when datepart(month, dateadd(month, @offset, @date)) = 12 and datepart(day,
dateadd(month, @offset, @date)) > @monthEnd then -11
        when datepart(day, dateadd(month, @offset, @date)) > @monthEnd then 1
        else 0
    end as Fin_Month) t
go

```

The functions input variables are:

@date This is the date that we are using derive the Year, Quarter and Month - e.g. `select * from [dbo].[FinancialYQM]('2021-04-01', 0, 31)`.

@offset The 'offset' allows us derive the correct values if the beginning of the Financial year is not January - e.g. if the Financial year starts in April we need to use an offset of -3 e.g. `select * from [dbo].[FinancialYQM]('2021-04-01', -3, 31)`.

@monthEnd The last parameter is used if the 'Financial' month finishes earlier than the end of the month - e.g. `select * from [dbo].[FinancialYQM]('2021-04-01', -3, 25)` if it finishes on the 25th of the month.

This function can be used directly in CRM's views: e.g.

```

select Opportunity.*, Fin_Year, Fin_Quarter, Fin_Month, Fin_Date
from Opportunity
cross apply [dbo].[FinancialYQM](Oppo_Opened, 0, 31)

```

or we can use it within a Table Script to extract the values to store directly in our own custom fields.